
The Model

% Single Open Economy Model, Satellite Module to GPM- Iris version
!variables:transition

LGDP, LGDP_BAR, G, Y, GROWTH, GROWTH4, GROWTH_BAR, GROWTH4_BAR
UNR, UNR_GAP, UNR_BAR, UNR_G,
PIE, PIE4, LCPI, E4_PIE4, PIETAR, RS, RR, RR_BAR, E1_Y, LZ, LZ_BAR, LZ_GAP
YF, RR_US, RR_BAR_US, LZ_E, E1_PIE

!parameters

alpha1, alpha2, alpha3, alpha4
beta1, beta2, beta3, beta4, beta5,
gamma1, gamma2, gamma4,
lambda1, lambda2, lambda3,
phi, rho, tau, growth_ss, rr_bar_ss,
rhobar_us, rho_us, betaf, rr_bar_us_ss, unr_ss, pietar_ss

!shocks:transition

RES_LGDP_BAR, RES_Y, RES_G,
RES_UNR_GAP, RES_UNR_BAR, RES_UNR_G,
RES_PIE, RES_PIETAR, RES_RS, RES_RR_BAR, RES_RR_DIFF, RES_LZ_BAR,
RES_RR_BAR_US, RES_RR_US, RES_YF

!equations:transition

% Host

Y = LGDP - LGDP_BAR;

LGDP_BAR = LGDP_BAR{-1} + G/4 + RES_LGDP_BAR;
GROWTH = 4*(LGDP - LGDP{-1});
GROWTH4 = LGDP - LGDP{-4};
GROWTH_BAR = 4*(LGDP_BAR - LGDP_BAR{-1});
GROWTH4_BAR = LGDP_BAR - LGDP_BAR{-4};

G = tau*growth_ss + (1-tau)*G{-1} + RES_G;

Y = beta1*Y{-1} + beta2*Y{+1} - beta3*(RR{-1} - RR_BAR{-1}) + beta4*LZ_GAP{-1} +beta5*YF + RES_Y;

UNR_GAP = alpha1*UNR_GAP{-1} + alpha2*Y + RES_UNR_GAP;

UNR = UNR_BAR - UNR_GAP;

UNR_BAR = (1-alpha3)*UNR_BAR{-1} + alpha3*unr_ss + UNR_G + RES_UNR_BAR;
UNR_G = (1 - alpha4)*UNR_G{-1} + RES_UNR_G;

PIE = lambda1*PIE{+1} + (1-lambda1)*PIE{-1} + lambda2*Y{-1} + lambda3*(LZ-LZ{-1}) + RES_PIE;

PIE4 = (PIE + PIE{-1} + PIE{-2} + PIE{-3})/4;

LCPI = LCPI{-1} + PIE/4;

E4_PIE4 = PIE4{+4};

RR = RS - PIE{+1};

E1_PIE = PIE{+1};

RR_BAR = rho*rr_bar_ss + (1-rho)*RR_BAR{-1} + RES_RR_BAR;

RS = gamma1*RS{-1} + (1-gamma1)*(RR_BAR + E4_PIE4 + gamma2*(PIE4{+3} - PIETAR{+3}) + gamma4*Y) + RES_RS;

E1_Y = Y{+1};

PIETAR = PIETAR{-1} + RES_PIETAR;

LZ_GAP = LZ - LZ_BAR;

LZ_BAR = LZ_BAR{-1} + RES_LZ_BAR;

RR - RR_US = 4*(LZ_E-LZ) + RR_BAR - RR_BAR_US + RES_RR_DIFF;

LZ_E = phi*LZ{+1} + (1-phi)*LZ{-1};

%Foreign block

RR_BAR_US = rhobar_us*rr_bar_us_ss + (1-rhobar_us)*RR_BAR_US{-1} + RES_RR_BAR_US;

RR_US = rho_us*rr_bar_us_ss + (1-rho_us)*RR_US{-1} + RES_RR_US;

YF = betaf*YF{-1} + RES_YF;

!variables:measurement

LGDP_, LCPI_, UNR_, RS_, LZ_, RR_US_,RR_BAR_US_, YF_

!equations:measurement

LGDP_ = LGDP;
LCPI_ = LCPI;
UNR_ = UNR;
RS_ = RS;
LZ_ = LZ;
RR_US_ = RR_US;

RR_BAR_US_ = RR_BAR_US;
YF_=YF;

Regularized Maximum Likelihood

Parameter Prior Posterior
Mode Dispersion Mode Dispersion

alpha1 0.650 0.158 0.651 0.020
alpha2 0.050 0.253 0.051 0.034
alpha3 0.200 0.158 0.202 0.020
alpha4 0.500 0.316 0.508 0.039
beta1 0.500 0.316 0.515 0.036
beta2 0.250 0.158 0.255 0.020
beta3 0.150 0.221 0.149 0.027
beta4 0.050 0.009 0.050 0.001
beta5 0.200 0.158 0.207 0.020
betaf 0.750 0.316 0.794 0.019
gamma1 0.700 0.158 0.715 0.079
gamma2 1.050 0.632 1.035 0.020
gamma4 0.150 0.158 0.146 0.034
lambda1 0.550 0.253 0.617 0.027
lambda2 0.100 0.316 0.053 0.020
lambda3 0.100 0.158 0.090 0.065
phi 0.800 0.632 0.697 0.008
rho 0.050 0.063 0.052 0.008
tau 0.030 0.063 0.030 0.035

Parameter Prior Posterior
Mode Dispersion Mode Dispersion

std RES LGDP BAR 0.100 0.006 0.370 0.020
std RES Y 0.400 0.632 1.101 0.034
std RES G 0.050 0.009 0.185 0.020
std RES UNR GAP 0.600 0.632 1.302 0.039
std RES UNR BAR 0.500 0.632 1.111 0.036
std RES UNR G 0.500 0.632 1.029 0.020
std RES PIE 1.000 0.632 2.893 0.027
std RES RR BAR 0.300 0.063 1.107 0.001
std RES RS 0.200 0.632 0.774 0.020
std RES PIETAR 0.100 0.632 0.281 0.019
std RES RR DIFF 2.000 0.632 7.337 0.079
std RES LZ BAR 2.000 0.632 7.338 0.020
std RES YF 0.400 0.632 0.719 0.034
std RES RR BAR US 0.300 0.063 1.162 0.027
std RES RR US 0.250 0.632 1.627 0.020

Calibrated Parameters

Parameter Value
growth ss 3.000
rr bar ss 2.500
rhobar us 0.900
rho us 0.800
rr bar us ss 1.900
unr ss 4.000
pietar ss 3.000

RMSEs

Parameter 1Q Ahead 4Q Ahead 8Q Ahead
GROWTH 6.981 7.146 7.169
GROWTH4 1.745 4.086 4.006
RS 0.535 1.731 1.839
PIE 1.717 1.695 1.438
PIE4 0.429 1.119 0.879

Mexico: Economic activity gaps

Output Gap +/- 2 std devs (w/o model or parameter uncertainty)

Output Growth (Y/Y) +/- 2 std devs (w/o model or parameter uncertainty)

